Листинг 8.4. Программа пересчета денежных сумм

unit uExt;
interface
uses
  Windows, Messages, SysUtils, Classes, Graphics,

  Controls, Forms, Dialogs, StdCtrls, Grids;
type  TForm1 = class(TForm)
    btnCalculation: TButton;
           lblDate: TLabel;
           btnExit: TButton;
        sgCurrency: TStringGrid;
         edtAmount: TEdit;
         lblAmount: TLabel;
           lblRate: TLabel;
           edtRate: TEdit;
       lblCurrency: TLabel;
            btnAdd: TButton;
         btnDelete: TButton;
    procedure btnExitClick(Sender: TObject);
    procedure FormCreate(Sender: TObject);
    procedure btnCalculationClick(Sender: TObject);
    procedure edtAmountKeyPress(Sender: TObject; var Key: Char);
    procedure edtRateKeyPress(Sender: TObject; var Key: Char);
    procedure btnAddClick(Sender: TObject);
    procedure btnDeleteClick(Sender: TObject);
  private
    { Private declarations }
  public
    { Public declarations }
  end;
var
  Form1: TForm1;
implementation
{$R *.DFM}
// Большинство действий, заданных в процедуре,
// можно выполнить через Инспектор объектов
procedure TForm1.FormCreate(Sender: TObject);
begin
// Установка свойств таблицы
sgCurrency.ColCount := 3;
sgCurrency.RowCount := 5;
sgCurrency.FixedCols := 0;
sgCurrency.FixedRows := 1;
sgCurrency.Cells[0,0] := 'Валюта';
sgCurrency.Cells[1,0] := 'Курс к доллару';
sgCurrency.Cells[2,0] := 'Сумма';
sgCurrency.ColWidths[0] := 160;
sgCurrency.ColWidths[1] := 90;
sgCurrency.ColWidths[2] := 120;
sgCurrency.Cells[0,1] := 'Американский доллар';
sgCurrency.Cells[0,2] := 'Евро';
sgCurrency.Cells[0,3] := 'Финская марка';
sgCurrency.Cells[0,4] := 'Английский фунт стерлингов';
sgCurrency.Cells[1,1] := '1';
sgCurrency.Cells[1,2] := '2' + DecimalSeparator + '18';
sgCurrency.Cells[1,3] := '6' + DecimalSeparator + '63';
sgCurrency.Cells[1,4] := '0' + DecimalSeparator + '69';
sgCurrency.Options := [goFixedVertLine, goFixedHorzLine, goVertLine,

     goHorzLine, goRangeSelect, goAlwaysShowEditor, goEditing];
// Установка свойств других элементов интерфейса
lblDate.Caption := 'По состоянию на ' + DateToStr(Date);
lblAmount.Caption := '&Сумма в рублях';
lblAmount.FocusControl := edtAmount;
edtAmount.Text := '';
lblRate.Caption := '&Курс рубля к доллару';
lblRate.FocusControl := edtRate;
edtRate.Text := '28' + DecimalSeparator + '86';
lblCurrency.Caption := 'Сумма в &Валюте';
lblCurrency.FocusControl := sgCurrency;
btnCalculation.Default := True;
end;
procedure TForm1.btnCalculationClick(Sender: TObject);
var                                    n: longint;
    Amount, Currency, Rate, CurrencyRate: real;
begin
// Преобразование суммы и курса в вещественные числа
// В случае ошибки - выдача сообщения и выход из процедуры
try
   Amount := StrToFloat(edtAmount.Text);
   Rate := StrToFloat(edtRate.Text);
   except
   MessageDlg('Ошибка ввода начальных данных!', mtError, [mbOK], 0);
   edtAmount.SetFocus;
   Raise EAbort.Create('');
   end;
// Расчет значения суммы для каждой строки таблицы
// и запись этого значения в соответствующую ячейку.
// В процессе вычислений обрабатываются возможные ошибки
for n := 1 to sgCurrency.RowCount - 1 do begin
    try
       CurrencyRate := StrToFloat(sgCurrency.Cells[1,n]);
       Currency := Amount / Rate * CurrencyRate;
       sgCurrency.Cells[2,n] := FloatToStrF(Currency, ffFixed, 10, 2);
       except
       MessageDlg('Ошибка расчета!', mtError, [mbOK], 0);
       sgCurrency.Cells[2,n] := '?';
       end;
    end;
end;
procedure TForm1.edtAmountKeyPress(Sender: TObject; var Key: Char);
begin
// Вводить только символы, допустимые для вещественного числа,
// и обрабатывать нажатия клавиши <BackSpace>
if not (Key in [#8,'0'..'9','+','E','e',DecimalSeparator]) then Key := #0;
end;
procedure TForm1.edtRateKeyPress(Sender: TObject; var Key: Char);
begin
// Вводить только символы, допустимые для вещественного числа,
// и обрабатывать нажатия клавиши <BackSpace>
if not (Key in [#8,'0'..'9','+','E','e',DecimalSeparator]) then Key := #0;
end;
procedure TForm1.btnAddClick(Sender: TObject);
begin
// Добавление новой строки в конец таблицы
sgCurrency.RowCount := sgCurrency.RowCount + 1;
// Очистка ячеек добавленной строки
sgCurrency.Rows[sgCurrency.RowCount - 1].Clear;
end;
procedure TForm1.btnDeleteClick(Sender: TObject);
var n :longint;
begin
// Если осталось две строки, операцию удаления не выполнять
if sgCurrency.RowCount = 2 then exit;
// Сдвиг строк вверх, начиная со строки, содержащей выделенную

for n := sgCurrency.Row to sgCurrency.RowCount - 2 do
    sgCurrency.Rows[n] := sgCurrency.Rows[n + 1];
// Удаление последней строки таблицы
sgCurrency.RowCount := sgCurrency.RowCount - 1;
end;
procedure TForm1.btnExitClick(Sender: TObject);
begin
Close;
end;
end.
