Листинг 19.6. Пример формы приложения для работы с товарами

unit uChange;
interface
uses
  Windows, Messages, SysUtils, Classes, Graphics, Controls, Forms, 
     Dialogs, Db, DBTables, Grids, DBGrids, StdCtrls, Mask, 
     DBCtrls, ExtCtrls;
type
  TForm1 = class(TForm)
        DBGrid1: TDBGrid;
    DataSource1: TDataSource;
         Table1: TTable;
        btnEdit: TButton;
      cbCanEdit: TCheckBox;
      pnlChange: TPanel;
         Label1: TLabel;
        dbeName: TDBEdit;
       dbePrice: TDBEdit;
        dbeNote: TDBEdit;
         Label2: TLabel;
         Label3: TLabel;
         Label4: TLabel;
  lblChangeKind: TLabel;
    btnChangeOK: TButton;
btnChangeCancel: TButton;
       dbcbUnit: TDBComboBox;
       btnClose: TButton;
 lblRecordCount: TLabel;
      btnInsert: TButton;
      btnDelete: TButton;
    procedure StateBrowse(Sender: TObject);
    procedure StateChange(Sender: TObject);
    procedure btnEditClick(Sender: TObject);
    procedure cbCanEditClick(Sender: TObject);
    procedure Table1BeforeEdit(DataSet: TDataSet);
    procedure btnCloseClick(Sender: TObject);
    procedure FormCreate(Sender: TObject);
    procedure btnChangeOKClick(Sender: TObject);
    procedure btnChangeCancelClick(Sender: TObject);
    procedure btnDeleteClick(Sender: TObject);
    procedure btnInsertClick(Sender: TObject);
    procedure Table1BeforePost(DataSet: TDataSet);
    procedure Table1BeforeInsert(DataSet: TDataSet);
  private
    { Private declarations }
  public
    { Public declarations }
  end;
var
  Form1: TForm1;
implementation
{$R *.DFM}
// Режим просмотра
procedure TForm1.StateBrowse(Sender: TObject);
begin
lblRecordCount.Caption := 'Всего записей — ' + IntToStr(Table1.RecordCount);
cbCanEditClick(Sender);
btnChangeOK.Enabled      := False;
btnChangeCancel.Enabled  := False;
lblChangeKind.Font.Color := clBlack;
lblChangeKind.Caption    := 'ПРОСМОТР ЗАПИСИ';
end;
// Режимы изменения
procedure TForm1.StateChange(Sender: TObject);
begin
btnEdit.Enabled         := False;
btnInsert.Enabled       := False;
btnDelete.Enabled       := False;
btnChangeOK.Enabled     := True;
btnChangeCancel.Enabled := True;
end;
procedure TForm1.FormCreate(Sender: TObject);
begin
// Исходное состояние элементов управления
StateBrowse(Sender);
// Первоначально изменение записей запрещено
cbCanEdit.Checked := False;
// Запрет автоматического перехода в режим редактирования
DataSource1.AutoEdit := False;
// Формирование списка единиц измерения товара
dbcbUnit.Items.Clear;
dbcbUnit.Items.Add('шт.');
dbcbUnit.Items.Add('упак.');
dbcbUnit.Items.Add('кор.');
end;
procedure TForm1.cbCanEditClick(Sender: TObject);
var bm1: TBookmark;
begin
// Запоминание положения текущей записи
bm1 := Table1.GetBookmark;
// Отключение отображения изменений данных в визуальных компонентах
Table1.DisableControls;
if not cbCanEdit.Checked then begin
   Table1.Active   := False;
   Table1.ReadOnly := True;
   Table1.Active   := True;
   // Блокирование элементов, связанных с переходом

   // в режимы изменения записей
   btnEdit.Enabled   := False;
   btnInsert.Enabled := False;
   btnDelete.Enabled := False;
   end
   else begin
   Table1.Active   := False;
   Table1.ReadOnly := False;
   Table1.Active   := True;
   // Разблокирование элементов, связанных с переходом

   // в режимы изменения записей
   btnEdit.Enabled   := True;
   btnInsert.Enabled := True;
   // Если набор данных пуст, то удаление записей запрещено
   if Table1.RecordCount > 0
      then btnDelete.Enabled := True
      else btnDelete.Enabled := False;
   end;
// Возврат к текущей записи
if Table1.BookmarkValid(bm1) then Table1.GotoBookmark(bm1);
if Table1.BookmarkValid(bm1) then Table1.FreeBookmark(bm1);
// Включение отображения изменений данных в визуальных компонентах
Table1.EnableControls;
end;
procedure TForm1.Table1BeforeEdit(DataSet: TDataSet);
begin
// Запрос на подтверждение перехода в режим редактирования
if MessageDlg('Будете редактировать?',
       mtConfirmation, [mbYes, mbNo], 0) <> mrYes then Abort;
end;
// Переход в режим редактирования
procedure TForm1.btnEditClick(Sender: TObject);
begin
Table1.Edit;
lblChangeKind.Font.Color := clRed;
lblChangeKind.Caption    := 'РЕДАКТИРОВАНИЕ ЗАПИСИ';
StateChange(Sender);
if dbeName.CanFocus then dbeName.SetFocus;
end;
procedure TForm1.Table1BeforeInsert(DataSet: TDataSet);
begin
// Подтверждение перехода в режим вставки
if MessageDlg('Добавить запись?',
       mtConfirmation, [mbYes, mbNo], 0) <> mrYes then Abort;
end;
// Переход в режим вставки
procedure TForm1.btnInsertClick(Sender: TObject);
begin
Table1.Insert;
lblChangeKind.Font.Color := clBlue;
lblChangeKind.Caption    := 'ВСТАВКА ЗАПИСИ';
StateChange(Sender);
if dbeName.CanFocus then dbeName.SetFocus;
end;
// Переход в режим просмотра удаляемой записи
procedure TForm1.btnDeleteClick(Sender: TObject);
begin
// Подтверждение перехода в режим просмотра удаляемой записи
if MessageDlg('Удалить запись?',
       mtConfirmation, [mbYes, mbNo], 0) <> mrYes then Exit;
lblChangeKind.Font.Color := clRed;
lblChangeKind.Caption    := 'УДАЛЕНИЕ ЗАПИСИ';
StateChange(Sender);
if btnChangeCancel.CanFocus then btnChangeCancel.SetFocus;
end;
procedure TForm1.Table1BeforePost(DataSet: TDataSet);
begin
// Проверка, задано ли значение для названия товара
if dbeName.Text = '' then begin
   Beep;
   MessageDlg('Не задано название товара!', mtError, [mbOK], 0);
   if dbeName.CanFocus then dbeName.SetFocus;
   Abort;
   end;
// Проверка, задано ли значение цены товара
if dbePrice.Text = '' then begin
   Beep;
   MessageDlg('Не задана цена товара!', mtError, [mbOK], 0);
   if dbePrice.CanFocus then dbePrice.SetFocus;
   Abort;
   end;
end;
procedure TForm1.btnChangeOKClick(Sender: TObject);
begin
// Утверждение изменений в текущей записи (редактируемой или новой)
// или удаления текущей записи (просматриваемой)
if Table1.State in [dsEdit, dsInsert]
   then Table1.Post
   else if lblChangeKind.Caption = 'УДАЛЕНИЕ ЗАПИСИ' then Table1.Delete;
StateBrowse(Sender);
end;
procedure TForm1.btnChangeCancelClick(Sender: TObject);
begin
// Если набор данных находился в режиме просмотра (при удалении записи),
// то никаких действий метод Cancel не выполняет
Table1.Cancel;
StateBrowse(Sender);
end;
// Закрытие формы 
procedure TForm1.btnCloseClick(Sender: TObject);
begin
Close;
end;
end.
