Листинг 34.1. Пример отправки почтового сообщения с использованием MAPI.
unit Unit1;

interface

uses

  Windows, Messages, SysUtils, Classes, Graphics, 

  Controls, Forms, Dialogs, StdCtrls;

type

  TForm1 = class(TForm)

        btnSend: TButton;

    edtReceiver: TEdit;

       edtTitle: TEdit;

         Label1: TLabel;

         Label3: TLabel;

  edtAttachment: TEdit;

 memMessageText: TMemo;

         Label4: TLabel;

         Label5: TLabel;

  btnInsertFile: TButton;

    OpenDialog1: TOpenDialog;

    procedure btnSendClick(Sender: TObject);

    procedure btnInsertFileClick(Sender: TObject);

  private

    { Private declarations }

  public

    { Public declarations }

  end;

var

  Form1: TForm1;

implementation
// Этот модуль обеспечивает возможность работы с электронной почтой

uses Mapi;

{$R *.DFM}

// Подготовить и отправить почтовое сообщение

function SendEmail(Receiver, Title, MessageText, 

                   AttachmentPath: string): boolean;

var MapiMessage: TMapiMessage;

         MError: Cardinal;

      Recipient: TMapiRecipDesc;

File_Attachment: TMapiFileDesc;

begin
// Подготовить информацию получателя

FillChar(Recipient, SizeOf(Recipient), 0);

Recipient.ulRecipClass := MAPI_TO;

Recipient.lpszName := PChar(Receiver);

// Подготовить информацию прикрепленного файла

if Trim(AttachmentPath) <> '' then begin

   FillChar(File_Attachment, SizeOf(File_Attachment), 0);

   File_Attachment.nPosition := ULONG(-1);

   File_Attachment.lpszPathName := PChar(AttachmentPath);

   end;

// Сформировать сообщение

with MapiMessage do begin

    ulReserved := 0;

    lpszSubject := PChar(Title);

    lpszNoteText := PChar(MessageText);

    lpszMessageType := nil;

    lpszDateReceived := nil;

    lpszConversationID := nil;

    flFlags := 0;

    lpOriginator := nil;

    nRecipCount := 1;

    lpRecips := @Recipient;

    if Trim(AttachmentPath) <> '' then begin

       nFileCount := 1;

       lpFiles := @File_Attachment;

       end

    else begin

       nFileCount := 0;

       lpFiles := nil;

       end;

  end;

// Попробовать отправить сообщение

MError := MapiSendMail(0, 0, MapiMessage, MAPI_DIALOG, 0);

if not (MError = SUCCESS_SUCCESS) then begin

   ShowMessage('Сообщение не было отправлено!');

   Result := False;

   end

   else Result := True;

end;

// Отправить сообщение

procedure TForm1.btnSendClick(Sender: TObject);

begin

SendEmail(edtReceiver.Text, edtTitle.Text,

         memMessageText.Text, edtAttachment.Text);

end;

// Выбрать прикрепляемый файл

procedure TForm1.btnInsertFileClick(Sender: TObject);

begin

if OpenDialog1.Execute then

   edtAttachment.Text := OpenDialog1.FileName;

end;

end.

